

A Microcomputer System with an Analogue/ Digital Converter for Cytophotometry*

Hiroyuki KURODA** and Ken SHIMAJI**

Abstracts—A computer program was designed for processing voltages which were generated in a photomultiplier attached to a microscope. The light intensities passed through sample specimens, i. e. voltage values, were digitized and converted to absorbances by this microscope-microcomputer system. This computer program was written in BASIC and consists of three subroutines, i. e. analogue/digital conversion, reading data into a diskette, and processing parts. The absorbances thus obtained were then summarized as frequency histograms with sample numbers, mean values, variances and standard deviations for every morphological cell types on CRT (Cathode Ray Tube). Possible applications and limitations of this system were described.

1. Introduction

Several microspectrophotometers have been introduced for analyzing wood elements. Some findings by the method throw a light on the distribution of lignins^{1,2)}, and extractives^{3,4)}. This method have also been intensively used for the measurement of DNA contents, or for the studies of ploidy levels in a tissue^{7~9)}. However, the equipments for these analyses are usually very expensive if provided in a full system.

We have been measuring spectra by a microscope equipped with a monochromator and a photomultiplier⁴⁾, although the procedure was very tedious and time-consuming. Our single beam system only outputs a linear voltage correlated to the transmittance of the light path. Although reference- and sample-voltages were recorded on a chart, the corresponding absorbances have to be calculated from both of the voltages one by one by a calculator through the wave-length subjected.

Recent microcomputers have been extremely developed in their performance and are available for many data processings⁵⁾ at a reasonable price. We are trying to develop a low-cost microspectrophotometer, because, in addition to the economical viewpoints, more accurate and rapid processings are required for the microscopic analyses, such as ploidy levels and cell contents in woody plants. Present paper deals a part of our development especially for a microcomputer system which was

* Partly supported by a Scientific Research Fund No. 58460220 of the Ministry of Education, Science and Culture.

** Research Section of Wood Biology.

introduced as a data processor for cytophotometry.

2. The Hardware System For Cytophotometry

Schematic representation of the system is shown in Figure 1. Figure 1A shows a microscope part. It consists of a Leitz UVMP microscope⁶⁾ equipped with a monochromator and a photomultiplier. A light source is commonly a Tungsten-halogen lamp (100W, 8A). A xenon lamp is also available as an alternative when the light intensity is too small to analyze. However, the xenon arc-instability results in invalid absorption data because of the single beam system. The specimen is illuminated by Köhler method, although the incident light is non-parallel. The photomultiplier is tensioned high voltages at a range from 0.65 to 1.10 KV.

Figure 1B is a data processing part, of which computer program will be described below. A linear voltage generated in the photomultiplier is converted to a digital signal by an A/D (Analogue/Digital) converter. The conversion circuit-board (API-2) was purchased from Contex Inc. and was inserted into a slot bus of a 8-bit microcomputer (NEC PC8001 Mark II). After the data processings, respective absorbances can be memorized in a 5-inch floppy diskette (disk unit: Epson TF20). All of the




Figure 1. Schematic representation of the microscope and data-processing units.

equipments in B are made in Japan and are obtainable at the price less than four hundred thousand yen.

The A/D conversion circuit-board is equipped with two input channels for analogue data, an A/D conversion element ADC80, PROM (Programable Read Only Memory) 8255 etc. An A/D conversion takes for 50 microseconds in this system. The ADC-80 outputs a digital signal of a 12 bit-TTL (Transistor-Transistor-Logic) level. The circuit was set a voltage-range from 0 to 5 volts for a input level, and thus one bit corresponds to 1.22 mV at the voltage-range. The circuit is available at a maximum voltage-range from -10 to 10 volts, although the accuracy is decreased (one bit corresponds to 4.88 mV). The resolution, i.e. voltages per one bit, can be theoretically improved by setting smaller range, but large signal/noise ratios from the photomultiplier make it ineffective.

3. A Computer Program For Cytophotometry

This program was designed for analyzing the absorbances of a cell or cell contents. Such analyses with or without computer have been used for the measurements of DNA contents or ploidy levels in nuclei, and brought some important findings in the cell cycle mechanisms especially in animal cells.^{7,9)} In plant tissues, O'Brien¹⁰⁾ and Dodds¹¹⁾ described on ploidy levels which may or may not control the differentiation of tracheary or xylem elements. King⁸⁾ also reviewed plant cell cycles, where the detailed methods were mentioned. Thus, the method may reveal some physiological roles more definitely in every morphological cell types, e.g. axial and ray parenchyma cells or big and small cells in either *in vivo* or *in vitro*. Chlorophyll contents or heartwood pigments in a plant cell may also become an object in microspectrophotometry.

In the analyses mentioned above, huge amounts of measurements and the data-processing are required. It is time-consuming and laborious work for the results obtained. The computer system extremely relieves the work load. This system displays every original data and processed results on a CRT (Cathode Ray Tube).

Every analogue data or voltages are first digitized and input to a computer. The digitized voltages are then assigned to either a sample or a reference. A signal from samples was classified into three morphological cell categories, i.e. in this case, ray parenchyma cells, fibers, and vessel elements. The digitized voltages are respectively memorized in a disk system through every three cell types. On the other hand, the voltages from a reference, i.e. no sample in the light path, are directly digitized and memorized in the same diskette.

Absorbances at respective spots on a specimen are calculated from corresponding sample voltages and a mean voltage of the reference, which were memorized in a diskette in advance. Finally, the absorbances at every morphological cell types are

processed to sample numbers, mean values, variances, standard deviations and frequency histograms of the absorbances.

This program was useful for analyzing spherical nuclei in order to evaluate ploidy levels in every morphological cell types during differentiation. However, the hardware system is insufficient if nuclei show irregular forms or if more accurate analyses are required. Because a field diaphragm and a measuring stop do not cut stray light which was passed through cell contents other than a nucleus subjected. In these cases, a nucleus have to be scanned by a small spot of incident light and respective intensities of the transmitted light have to be integrated through the nucleus. Computer controlled X-Y stage is often employed for this purpose but it costs expensive. Thus, as an alternative, we are trying to develop a system that area of a nucleus in a microscope field is able to measure by means of a drawing apparatus combined with a digitizer.

4. Practice of The Program

The program was written in N80 disk BASIC. This will be applicable to most of the microcomputers with a little modifications, because the language is originated from Microsoft BASIC system. The general flow chart of the program was shown in Figure 2.




Figure 2. Flow chart of a data-processings in a computer-program.

When the program are started after setting sample-specimen under the microscope, a message as shown in Figure 3a appears on the CRT. If the microscope view is focused on no sample area, select "1" and hit "RETURN" key on the key board for reference record. If the view is focused on the cell content subjected, e.g. a nucleus, then hit "2" and "RETURN" keys for sample record. If the additional data have to be recorded, select "3" or "4" according to reference or sample. Then, a message as shown in Figure 3b and 3c (or 3d) will appear in the respective cases of the reference (or sample). As shown in the message, set a diskette for data files on drive #2, then select a key "RETURN" for reference record or "f", "v", "r" for sample record.




Figure 3. Representative CRT displays of the instructions and results..

The analogue value generated in the photomultiplier is automatically digitized and displayed on a CRT as shown in Figure 3e. Where, S(number) shows sample number and "f", "v", "r" mean that fiber, vessel or ray cells were respectively selected. At the same time, the sample context displayed on a CRT is memorized one by one as a sequential file in the disk system. Then, the same message as shown in Figure 3b and 3c (or 3d) will appear again. The voltages are subsequently digitized and memorized as described above until the key "e" is selected. When the key "e" is hit, the message shown in Figure 3a will appear. If additional data are need to be memorized, the key "3" or "4" will be selected. If one want to read out the data memorized in the diskette, the key "5" will be selected.

When the key "5" is selected, Figure 3d will appear on the CRT. After the key selection, the CRT displays respective voltages, corresponding to the references,




Figure 4. Data summary on the CRT display.

samples and the mean of references (Figure 4). Then, the mean of references, respective voltages and the corresponding absorbances are displayed or printed out on a CRT or a printer as shown in Figure 4a. As a summary, frequency at a 0.2 absorbance unit, numbers of input error, net sample numbers, mean value, variance, and standard deviation will be displayed. The frequency histogram is also displayed as shown in Figure 4b.

5. Program List

```

10 REM *****
20 REM **** ANALOGUE/DIGITAL CONVERSION ****
30 REM **** FOR CYTOPHOTOMETRY ****
40 REM ****
50 REM **** by KURODA,H. Aug.,1983 ****
60 REM *****
70 CLEAR 1000:CMD CLS 3:CONSOLE,,0:WIDTH 36,25
80 DIM R(500),S(500),A(500)
90 REM ****key selection*****
100 LOCATE 22,0:PRINT "SELECT KEY NO";
110 LOCATE 22,2:PRINT "      ref  samp"
120 LOCATE 22,3:PRINT "      save  1  2"
130 LOCATE 22,4:PRINT "      add   3  4"
140 LOCATE 22,5:PRINT "      load  5  5"
150 LOCATE 22,6:PRINT "      end   6  6"
160 LOCATE 25,8:INPUT "key no=";KS
170 ON KS GOTO 250,420,400,580,760,190
180 IF KS<1 OR KS>7 THEN 90
190 CMD CLS 3:END
200 LOCATE 22,12:PRINT "SET A DISK"
210 LOCATE 23,13:PRINT "ON DRIVE #2 !"
220 LOCATE 21,18:PRINT "SELECT INITIAL:"
230 LOCATE 22,19:PRINT "f)iber,v)essel"
240 LOCATE 22,20:PRINT "r)ay or e)nd":RETURN
250 REM ****REF save/add*****
260 OPEN "2:REF" FOR OUTPUT AS #1
270 K=-1:FOR I=1 TO 999:K=K+1
280 LOCATE 22,12:PRINT "SET A DISK"
290 LOCATE 23,13:PRINT "ON DRIVE #2 !"
300 LOCATE 22,18:PRINT "SELECT INITIAL:"
310 LOCATE 22,19:PRINT "return or e)nd"
320 IN$=INKEY$
330 IF IN$="" THEN 320
340 IF IN$="e" THEN 190
350 GOSUB 620:R(I)=S(I):LOCATE 0,K
360 IF K=20 THEN K=0:CMD CLS
370 PRINT "R(" I")=";:PRINT USING "#.#### V";R(I)
380 PRINT #1, R(I);", ";
390 NEXT I:CLOSE
400 OPEN "2:REF" FOR APPEND AS #1
410 GOTO 270
420 REM ****SAMP save/add *****
430 OPEN "2:FBR" FOR OUTPUT AS #2
440 OPEN "2:VSL" FOR OUTPUT AS #3
450 OPEN "2:RAY" FOR OUTPUT AS #4
460 K=-1:FOR I=1 TO 999:K=K+1
470 IF K=20 THEN K=0:CMD CLS
480 GOSUB 620:GOSUB 220:LOCATE 0,
490 IN$=INKEY$
500 IF IN$="" THEN 490
510 IF IN$="f" THEN PRINT USING "f:S('##')=#.#### V";I;S(I):GOTO 550
520 IF IN$="v" THEN PRINT USING "v:S('##')=#.#### V";I;S(I):GOTO 560
530 IF IN$="r" THEN PRINT USING "r:S('##')=#.#### V";I;S(I):GOTO 570
540 IF IN$="e" THEN 190 ELSE 480
550 PRINT #2,S(I);", ";:NEXT I:GOTO 190
560 PRINT #3,S(I);", ";:NEXT I:GOTO 190
570 PRINT #4,S(I);", ";:NEXT I:GOTO 190
580 OPEN "2:FBR" FOR APPEND AS #2
590 OPEN "2:VSL" FOR APPEND AS #3
600 OPEN "2:RAY" FOR APPEND AS #4
610 GOTO 460
620 REM ****AD conversion, subroutine*****
630 OUT &H83,&H92 REM****8255 initialize
640 OUT &H82,&H2 REM****CH1 select
650 FOR A=1 TO 100:NEXT A REM****wait
660 OUT &H82,&H3 REM****CH1 start on
670 OUT &H82,&H2 REM****CH1 start off
680 M=INP(&H81) REM****ADC80 status

```

```

690 IF (M AND &H10)=1 THEN 680
700 B8T=INP(&H80) REM♦♦♦♦8bit data
710 B4T=INP(&H81) REM♦♦♦♦4bit data
720 B4T=B4T AND &HF
730 B12T=B8T+B4T*&H100 REM♦♦♦♦12bit data
740 S(I)=10-20*B12T/&HFFF REM♦♦♦♦voltage
750 RETURN
760 REM ♦♦♦♦♦REF load♦♦♦♦♦
770 OPEN "2:REF" FOR INPUT AS #1
780 K=-1:FOR I=1 TO 999:K=K+1
790 IF EOF(1) THEN 860
800 INPUT #1, R(I)
810 IF K=20 THEN K=0:CMD CLS
820 LOCATE 0,K
830 PRINT USING "R(##)=#.#### V";I;R(I)
840 IF R(I)<8E-03 THEN W=W+1:NEXT I
850 RT=RT+R(I):NEXT I
860 R=RT/(I-1-W)
870 PRINT:PRINT USING "N=## RT=###.### R=#.### V";I-1;RT;R
880 CLOSE
890 REM ♦♦♦♦♦SAMP load♦♦♦♦♦
900 LOCATE 3,22: PRINT "===SAMP LOAD:y)es or n)o?=== "
910 GOSUB 1810:CMD CLS:GOSUB 220:K=-1
920 LOCATE 0,K+1:IN$=INKEY$
930 IF IN$="" THEN 920
940 IF IN$="f" THEN PRINT "fiber":GOTO 980
950 IF IN$="v" THEN PRINT "vessel":GOTO 1040
960 IF IN$="r" THEN PRINT "ray":GOTO 1100
970 IF IN$="e" THEN 190 ELSE 910
980 OPEN "2:FBR" FOR INPUT AS #2
990 FOR I=1 TO 999:K=K+1
1000 IF EOF(2) THEN 1160
1010 INPUT #2,S(I)
1020 IF K=20 THEN K=0:CMD CLS
1030 PRINT USING "S(##)=#.#### V";I;S(I):NEXT I
1040 OPEN "2:VSL" FOR INPUT AS #3
1050 FOR I=1 TO 999:K=K+1
1060 IF EOF(3) THEN 1160
1070 INPUT #3,S(I)
1080 IF K=20 THEN K=0:CMD CLS
1090 PRINT USING "S(##)=#.#### V";I;S(I):NEXT I
1100 OPEN "2:RAY" FOR INPUT AS #4
1110 FOR I=1 TO 999:K=K+1
1120 IF EOF(4) THEN 1160
1130 INPUT #4,S(I)
1140 IF K=20 THEN K=0:CMD CLS
1150 PRINT USING "S(##)=#.#### V";I;S(I):NEXT I
1160 N=I-1:CLOSE
1170 REM ♦♦♦♦♦absorbance♦♦♦♦♦
1180 LOCATE 3,22
1190 PRINT "===ABSORBANCE y)es or n)o?=== "
1200 GOSUB 1810
1210 FOR I=1 TO N
1220 IF R<S(I) OR R<8E-03 OR S(I)<8E-03 THEN Z=Z+1:NEXT I
1230 A(I)=.434295*LOG(R/S(I))
1240 NEXT I
1250 REM ♦♦♦♦♦Data Print♦♦♦♦♦
1260 PRINT
1270 PRINT " no T-ref T-samp absorbance"
1280 FOR I=1 TO N
1290 PRINT USING "### ##.### ##.### ##.###";I;R;S(I);A(I)
1300 NEXT I
1310 REM ♦♦♦♦♦Frequency♦♦♦♦♦
1320 MAX=F(1)
1330 FOR I=1 TO N:FOR J=1 TO 9
1340 IF A(I)>(.2*J-.2) AND A(I)<=.2*J THEN F(J)=F(J)+1
1350 IF MAX<F(J) THEN MAX=F(J)
1360 NEXT J:NEXT I

```

```

1370 FOR I=1 TO N
1380 IF A(I)>1.8 THEN F(10)=F(10)+1
1390 IF MAX<F(10) THEN MAX=F(10)
1400 NEXT I:PRINT
1410 REM *****DATA SUMMARY*****
1420 PRINT "===SUMMARY y)es or n)o?===" :GOSUB 1810
1430 PRINT "frequency:" REM*****frequency
1440 FOR J=1 TO 10
1450 PRINT F(J);:NEXT J
1460 FOR I=1 TO N REM*****mean value
1470 T1=T1+A(I)
1480 NEXT I
1490 AM=T1/(N-Z)
1500 FOR I=1 TO N REM*****variance
1510 IF A(I)<1E-03 THEN NEXT I
1520 T2=T2+(A(I)-AM)^2
1530 NEXT I
1540 V=T2/(N-Z)
1550 S=SQR(V) REM*****standard deviation
1560 REM *****SUMMARY PRINT*****
1570 PRINT
1580 PRINT USING "error no. = ###";Z
1590 PRINT USING "sample no. = ###";N-Z
1600 PRINT USING "mean value = #.#####";AM
1610 PRINT USING "variance = #.#####";V
1620 PRINT USING "stand.dev. = #.#####";S
1630 REM *****GRAPHIC PATTERN*****
1640 PRINT
1650 PRINT "===GRAPH y)es/n)o? ===" :GOSUB 1810
1660 CMD SCREEN 1,0,3
1670 CMD CLS 3
1680 FOR J=1 TO 10
1690 A=32+24*(J-1)
1700 B=29+24*J
1710 C=176-168*F(J)/MAX
1720 CMD LINE (A,176)-(B,C),1,B
1730 NEXT J
1740 FOR Y=1 TO (MAX+1)
1750 UNT=176-168*(Y-1)/MAX
1760 CMD LINE (27,UNT)-(29,UNT)
1770 CMD LINE (272,UNT)-(274,UNT)
1780 NEXT Y
1790 GOSUB 1810
1800 END
1810 REM *****STOP SUBROUTINE*****
1820 IN$=INKEY$
1830 IF IN$="" THEN 1810
1840 IF IN$="y" THEN RETURN
1850 IF IN$="n" THEN CMD CLS ELSE 1820:END

```

References

- 1) GORING, D.A.I.: International Symposium on Wood & Pulping Chemistry, Stockholm, Session I:3-10 (1981).
- 2) SCOTT, J.A.N., PROCTER, A.R., FUGUS, B.J. and GORING, D.A.I.: Wood Sci. Technol., **3**, 73-92 (1969).
- 3) BLAND, D.E. and HILLIS, W.E.: Appita **23**, 204-210 (1969).
- 4) KURODA, H. and SHIMAJI, K.: IAWA Bull., **4**, 7 (1983); Holzforsch. **37**, 225-230 (1983).
- 5) NEC Inc.: PC8001 Mark II N80-BASIC Reference Manual (1983).
- 6) FREED, J.J.: in 'Physical Techniques in Biological Research 2nd ed., Vol. 3 Part C', Pollister, A.W. ed., Academic Press, pp 95-167 (1969).
- 7) JOHN, P.C.L. ed.: 'The Cell Cycle', Cambridge Univ. Press (1981).
- 8) KING, P.J.: in 'Advances in Biochemical Engineering Vol. 18' Fiechter, A. ed., Springer-Verlag, pp 1-38 (1980).
- 9) PATTISON, J.R., BITENSKY, L. and CHAYEN, J. eds.: 'Quantitative Cytochemistry and its

- Applications', Academic Press (1979).
- 10) O'BRIEN, T.P.: in 'Xylem Cell Development' Barnett, J.R. ed., Castel House Pub., p. 29 (1981).
 - 11) DODDS, J.H.: in 'Xylem Cell Development' Barnett J.R. ed., Castle House Pub., pp 153-167 (1981).